Tutorial 13: 
Subroutine Calls

New Instructions Introduced

CALL

RET

FRAME

MVMM

CALA

CC

Overview of Tutorial

It is normal programming practice to divide a large project into a series of smaller tasks. For this subroutines are used. This tutorial shows how to call subroutines and pass parameters to them. 

Subroutine Calls

Within the design of a program it is usually possible to break down the design into a series of smaller units known as subroutines. Example 13-1 shows a C program that has been divided into two subroutines named subroutine1() and subroutine2(): 

Example 13-1.

	

	void main(void)
	

	{
	

	  for ( ; ; ) 
	// Infinite loop.

	   {
	

	     subroutine1();
	// Call first subroutine.

	     subroutine2();
	// Call second subroutine.

	   }
	

	}
	


The code in Example 13-1 contains an infinite loop that repeatedly executes the subroutine subroutine1() followed by the subroutine subroutine2().
We shall now implement the code shown in Example 13-1 using assembly language. For this we shall use the instruction CALL (call unconditionally). The instruction CALL takes a single operand, the address in program memory where the body of the subroutine is to be found. Normally the operand takes the form of a label.

Example 13-2.

	main:
	CALL subroutine1
	; Call the subroutine to be found

; at the address given by the

; label subroutine1.

	
	CALL subroutine2
	; Call the subroutine to be found

; at the address given by the

; label subroutine2.

	
	B main
	; Loop back to the label main.

	
	
	

	subroutine1:
	
	; Body of subroutine goes here. 

	
	RET
	; Return from subroutine

	
	
	

	subroutine2:
	
	; Body of subroutine goes here.

	
	RET
	; Return from subroutine


During program execution, when the instruction CALL is encountered, some internal processing is done and a branch occurs to the address provided by the operand. Normally the code for the subroutine is put somewhere else in the program so as to make the main flow less cluttered.   

Example 13-2 shows two minimal subroutines named subroutine1 and subroutine2. The body of each subroutine is not shown. At the end of each subroutine, there must be the instruction RET (return from subroutine). The instruction RET causes execution to resume at the main program at the line following the instruction CALL.

If the instruction RET is inadvertently omitted, there will be no return to the calling subroutine and program execution will incorrectly continue at the line after the subroutine itself.     

The Mechanism of Subroutine Calls and Returns

When a subroutine call occurs, how does the program remember where to continue execution after the subroutine is completed?

Part of the internal processing at the time of executing the instruction CALL is to push the address of the instruction following the CALL instruction onto the stack. This address is referred to as the return address and is shown in Example 13-3. 

Example 13-3.

	
	
	

	3000
	CALL subroutine1
	; Call the subroutine at the  

	
	
	; program address which has the

; label subroutine1.

	3002
	ADD #5, A
	; The operation following

; immediately after the instruction

; CALL.


The left hand column of Example 13-3 shows the address in program memory of the code. The instruction CALL at program address 3000h takes two words of program memory. The instruction following immediately afterwards, here ADD #5, A, is  located at address 3002h. This is the return address.

When the instruction CALL subroutine1 is executed, the value 3002h is pushed onto the stack. When the instruction RET is reached, program execution continues at the address contained in the return address. The mechanism is shown in Figure 13-1.      

When the instruction CALL has been executed, the return address, here 3002h, is pushed onto the stack. 

At the end of a subroutine, the instruction RET pops the return address from the stack to the program counter (PC) and then increments the stack pointer. This is shown in Figure 13-2: 




After a call and a return, the stack pointer is restored to its original value. It is important that any stack operations within the subroutine such as PSHD, PSHM, POPD and POPM leave the stack pointer in the same condition it was when the subroutine was called. If not, some value other than the return address will be popped off the stack by the instruction RET.

Passing Parameters to a Subroutine   

Consider the simple case in C code where we wish to pass two parameters to a subroutine:

Example 13-4.

	

	int subroutine2(int a, int b);
	// Function prototype of a  

// subroutine that takes two 

// parameters. 

	subroutine2(0xAAAA, 0xBBBB);
	// Call made to subroutine2

// passing two parameters of

// value AAAAh and BBBBh.  

	subroutine2 (int a, int b)
	// Body of subroutine.

	{
	

	 return(a + b);
	// Add variables a and b.

	}
	


In assembly language, we have seen how to call a subroutine using the instruction CALL. How do we pass values to an assembly language subroutine? 

Several choices are available to us. We can put the parameters into accumulator A or accumulator B. 

Example 13-5.

	
	

	LD #0AAAAh, A
	; Load accumulator A with AAAAh.

	LD #0BBBBh, B
	; Load accumulator B with BBBBh.

	CALL subroutine2
	; Call the subroutine at program address 

; corresponding to the label subroutine2.


Within subroutine2, the two parameters a and b passed to the subroutine are available in accumulator A and accumulator B. 

Another way to pass the parameters is to use two of the auxiliary registers AR0 to AR7. 

Example 13-6.

	
	

	STM #0AAAAh, AR1
	; Load auxiliary register AR1 with AAAAh.

	STM #0BBBBh, AR2
	; Load auxiliary register AR2 with BBBBh.

	CALL subroutine2
	; Call the subroutine at program address 

; corresponding to the label subroutine2.


Within, subroutine2, parameter a is contained in AR1 and parameter b is contained in AR2.

Passing Parameters on the Stack

A different method that is widely used is to pass parameters on the stack. This has two advantages. First, a copy of each of the original values is passed to the subroutine, and therefore the originals remain unchanged. Second, using the stack allows us to pass a list of parameters. The length of the list can be variable.  

Let us now implement Example 13-4, this time passing parameters on the stack:

Example 13-7.

	
	

	LD #4, DP
	; Page 3. Gain access to data memory

; addresses 200h to 27Fh

	ST #0AAAAh, 64h
	; Store the value AAAAh at the data memory

; address 200h + 64h = 264h.

	ST #0BBBBh, 65h
	; Store the value BBBBh at the data memory

; address 200h + 65h = 265h. 

	PSHD 64h
	; Push parameter at data memory address 

; 200h + 64h onto the stack.

	PSHD 65h
	; Push parameter at data memory address 

; 200h + 65h onto the stack.

	CALL subroutine2
	; Call the subroutine at program address 

; corresponding to the label subroutine2.


When the code at subroutine2 is executed, parameters a and b are on the stack. 

The effect of the code in Example 13-7 upon the stack is shown in Figure 13-3.




The two parameters AAAAh and BBBBh are pushed onto the stack. The return address 3002h is also pushed onto the stack. 

Working on Variables Stored on the Stack

As part of the subroutine subroutine2, we shall want to gain access to the parameters passed on the stack. There is a slight problem here. After the subroutine call has been made, the stack pointer does not point to the last parameter passed on the stack, rather it returns to the return address. Before we can work on the parameters, some form of adjustment must be made to the stack pointer.

One way to work upon the parameters pushed onto the stack is to use indirect addressing.

This is shown in Example 13-8. 

Example 13-8.

	
	
	

	subroutine2: 
	MVMM SP, AR0
	; Copy stack pointer to auxiliary

; register AR0. AR0 points to the 

; return address. 

	
	ADDM #2, AR0
	; Increment AR0 to point to first

; parameter AAAAh.

	
	LD *AR0-, A
	; Carry out some operation using the

; first variable. Decrement AR0 to

; point to the second parameter; 

	
	ADD *AR0, A
	; Carry out some operation using the 

; second parameter BBBBh.  

	
	RET
	; Return to calling subroutine.

; The return value is now in

; accumulator A.


In Example 13-8 we have introduced the new instruction MVMM (move data from memory-mapped register to memory-mapped register). This takes two operands. The first operand 

must be an auxiliary register AR0 to AR7 or the stack pointer (SP). The second operand must also be an auxiliary register AR0 to AR7 or the stack pointer. This instruction allows us to copy values between auxiliary registers and the stack pointer. 

The first operation in Example 13-8 is to copy the stack pointer to AR0. This allows us to use indirect addressing. After the instruction MVMM, auxiliary register AR0 points to the same address in data memory as does the stack pointer, which is the return address. 

In order to work on the first parameter passed on the stack, we must increment the copy of the stack pointer by the number of parameters passed. Here we have incremented AR0 by two. We then carry out some operation using the parameters that have been passed. In this case we simply added them together. Finally we return from the subroutine using the instruction RET. Note that we do not need to decrement AR0 before returning because it has no effect on the stack pointer. The return value (AAAAh + BBBBh) is in accumulator A.     

By copying the stack pointer to an auxiliary register we can use indirect addressing to work on the values passed on the stack and leave the stack pointer unaffected. This is an inherently safe way of working because no correction to the stack pointer is required before the subroutine returns. 

Correcting the Stack Pointer after a Subroutine Call

When parameters are passed on the stack to a subroutine, the stack pointer needs to be corrected once the subroutine has been completed. Example 13-9 shows the code needed around a subroutine call. 

Example 13-9.

	
	
	

	 
	PSHD *AR2+
	; Push the first parameter onto the

; stack. Point to second parameter

	
	PSHD *AR2
	; Push the second parameter onto the

; stack.

	
	CALL subroutine2
	; Call to subroutine subroutine2. 

	
	POPD *AR2
	; Increment the stack pointer.

	
	POPD *AR2
	; Increment the stack pointer.


In Example 13-9 we push two parameters on the stack then call the subroutine subroutine2. This decrements the stack pointer by two. We then use two POPD instructions to increment the stack pointer to the value it was before the two values were passed on the stack.

Note that when parameters are passed on the stack to a subroutine, it is important to correct the stack pointer after the subroutine call. If not, two words of the stack will be lost (pushed but never popped) every time the subroutine is called. Eventually, there will be no more stack available.  

In Example 13-9 we have used the instruction POPD to correct the stack pointer. We could equally well have used the instruction POPM. 

Instead of using two POPD or POPM instructions we could also use the instruction FRAME (stack pointer immediate offset), as shown in Example 13-10:

Example 13-10.

	
	
	

	 
	PSHD *AR2+
	; Push the first variable onto the

; stack. Point to the next space on

; the stack.

	
	PSHD *AR2+
	; Push the second variable onto the

; stack.

	
	CALL subroutine2
	; Call to subroutine. 

	
	FRAME 2
	; After return from subroutine,

; increment the stack pointer by two

; to restore it to the value it was

; before the subroutine call.  


The instruction FRAME takes a single operand and this is an offset in the range -128 to +127. With this instruction it is possible to increment or decrement the stack without moving any data or using an auxiliary register. To increment the stack (similar to popping) we use a positive constant. To decrement the stack (similar to pushing) we use a negative constant. 

Creating a Stack Frame    

The C language makes use of local variables. These variables exist only while the subroutine is active and the memory allocated is freed for general purpose when the subroutine returns.

Example 13-11.

	
	

	void subroutine3(void)
	// Subroutine

	{
	

	 unsigned int a; 
	// Local variable. One word.

	 unsigned long b;
	// Local variable. Two words.

	 ...
	// Body of code.

	}
	


Within subroutine3, there are two local variables which require a total of three words of data memory. If these variables are on the stack, then the instruction FRAME can be used to create what is known as a stack frame. This is an area of the stack temporarily assigned for the storage of local variables, which exists only for the duration of the subroutine.  

Example 13-12.

	
	
	

	subroutine3: 
	FRAME -3
	; Decrement the stack pointer by 3.

; Make space for 3 words. 

	
	
	; Body of subroutine goes here.

	
	FRAME +3
	; Increment the stack pointer by 3. 

; Correct the stack pointer before

; returning to calling subroutine.

; Increment the stack pointer by 3.

	
	RET
	; Return to calling subroutine.


At the beginning of the subroutine, we make room on the stack for three local variables using the instruction FRAME with a negative constant. 

To gain access to the local variables in the stack frame, we can use indirect addressing in a similar way to that given in Example 13-8.

At the end of the subroutine the three words allocated for local variables are given back to the stack by executing the instruction FRAME with a positive operand. 

Conditional Subroutine Calls

There are cases when we wish to call a subroutine upon the outcome of a test. In Example 13-11, the subroutine subroutine4() is only called if the variable x is 5. 

Example 13-13.

	
	

	unsigned int x;
	// Variable.

	if ( x == 5) 
	// Test variable. 

	{ 
	

	  subroutine4();

}
	// Call subroutine when x

// is equal to 5. 


To implement Example 13-11 in assembly language, assuming that the variable x is stored at data memory address 460h then:

Example 13-14.

	
	

	LD #8, DP 
	; Page 8. Gain access to data memory

; addresses 400h to 47Fh.

	LD 60h, A
	; Copy variable x into accumulator A.

	XOR #5, A 
	; Compare value in accumulator A with the

; constant 5. 

	BC ANEQ, $+2
	; If accumulator A does not contain 0

; (the condition ANEQ evaluates to TRUE) then

; skip the subroutine call. 

	CALL subroutine4
	; Call subroutine.  


In Example 13-14, the instruction BC requires two words of program memory so that the address of the next instruction is at program counter + 2. Here we have used the symbol $+2 to indicate the current program address + 2. This saves the need to use a label.  

For cases like this, the TMS320C54x provides the special instruction CC (call conditionally). The instruction CC takes two operands. The first operand is the address in program memory of the subroutine to be called. The second operand is the test to be carried out. If the condition evaluates to TRUE, then the subroutine is called. This is shown in Example 13-5:  

Example 13-15.

	
	

	LD #8, DP 
	; Page 8. Gain access to data memory

; addresses 400h to 47Fh.

	LD 60h, A
	; Copy variable x into accumulator A.

	XOR #5, A 
	; Compare value in accumulator A with the

; constant 5. 

	CC subroutine4, AEQ 
	; If accumulator A contains zero (the

; condition AEQ evaluates to TRUE) then

; call the subroutine at the address

; given by the label subroutine4.


The instruction CC combines together the instruction BC (branch conditionally) and the instruction CALL (call unconditionally). The range of tests that can be used with the conditional call are the same as those available for the conditional branch BC. Using the instruction CC we may therefore test the following: Carry (C) flag set or clear, Overflow flags OVA or OVB, the test control flag (TC) or numeric comparisons with values in accumulators A or B.      

Let us take a second example. Say we wish to call a subroutine if the result of a multiplication has not produced an overflow, as shown in Example 13-16.

Example 13-16.

	
	

	RSBX OVM
	; Turn off overflow mode to allow

; overflows to occur.

	MPY *AR2, #2000, B
	; Multiply the value in the data memory

; address pointed to by AR2 by 2000 and

; put the product in accumulator B.

	CC subroutine5, BNOV 
	; If there has been no overflow from

; accumulator B (the condition BNOV

; evaluates to TRUE) then call the

; subroutine at the address identified by

; the label subroutine5.


Calculated Subroutine Calls

There are cases where we wish to call a subroutine dependant upon the value of a variable. 

When programming in C, we can do this using a series of if-else statements or with a switch statement as shown in Example 13-17. 

Example 13-17.

	

	unsigned int selector;
	// Controlling variable.

	switch(selector)
	

	{
	

	 case 0: subroutine1(); 
	// Selector is 0. Call

	         break;
	// subroutine1();

	 case 1: subroutine2();
	// Selector is 1. Call

	         break; 
	// subroutine2();

	 case 2: subroutine3(); 
	// Selector is 2. Call 

	         break;  
	// subroutine3().

	 case 3: subroutine4();
	// Selector is 3. Call

	         break;
	// subroutine4().

	}
	


This can be implemented using the instruction CALA (call subroutine at location specified by accumulator A). Assuming the variable selector is stored at data memory address 202h: 

Example 13-18 .

	
	
	

	
	LD #4, DP
	; Page 4. Gain access to data

; memory addresses 200h to

; 27Fh.

	
	LD  2h, A
	; Load contents of data

; memory address 202h into

; accumulator A.

	
	SFTA 1
	; Multiply the value in

; accumulator A by 2. (Each

; subroutine call takes 2

; words of program memory).

	
	ADD #subroutine1, A
	; Add starting address of

; subroutines.

	
	CALA
	; Call the subroutine at the

; address contained in

; accumulator A. 

	
	B done
	; Skip subroutines.

	subroutine1:
	B body_subroutine1   
	; 2 words.

	subroutine2:
	B body_subroutine2
	; 2 words.

	subroutine3:
	B body_subroutine3
	; 2 words.

	subroutine4: 
	B body_subroutine4
	; 2 words.

	done:
	
	


To better explain how the code in Example 13-18 works, the program memory addresses have been written in Example 13-19. 

Example 13-19.

	
	
	

	Program Memory Address
	
	

	0555
	
	LD #4, DP

	0556
	
	LD  2h, A ; selector

	0557
	
	SFTA 1    ; multiply by 2.

	0558
	
	ADD #subroutine1, A

	055A
	
	CALA

	055B
	
	B done

	055D
	subroutine1:
	B body_subroutine1   

	055F
	subroutine2:
	B body_subroutine2

	0561
	subroutine3:
	B body_subroutine3

	0563
	subroutine4: 
	B body_subroutine4

	0565
	done:
	


This instruction CALA works in a similar way to a look-up table. We load accumulator A with the starting address of the subroutines starting at the address subroutine1 (here 055Dh). Then we add the offset provided by the variable to accumulator A. Note that because each branch instruction takes 2 words of data memory, we must multiply the offset by 2.  

In Examples 13-18 and 13-19, the four subroutines are at addresses body_subroutine1, body_subroutine2, body_subroutine3 and body_subroutine4 respectively. Each subroutine must contain the instruction RET.  

Using a calculated call to a subroutine offers two advantages over a series of conditional calls. First, the calculated call requires less code. Second, with a series of conditional calls (if-else) there will be more instructions to be executed to reach the final subroutine than will be the case with the first. The code using the switch statement will always take the same number of instructions. 

Upgrading from the TMS320C2000 to the TMS320C5000 

All the devices support the instructions CALL, RET, CALA and CC.  

Table 13-1. Comparison of Instructions

	Description
	TMS320C2000

Instruction
	TMS320C5000

Instruction

	Call unconditionally
	CALL
	CALL

	Return from subroutine
	RET
	RET

	Call subroutine at location specified by accumulator
	CALA
	CALA

	Call conditionally
	CC
	CC


Questions

	1.
	What is meant by the expression subroutine call?

	2.
	Which one of the following is correct:

a) CALL

b) CALL A

c) CALL 3000h

d) CALL *AR2
e) CALL *AR0-

f) CALL subroutine4?

	3.
	Which one of the following is correct:

a) RET

b)RET A

c) RET 16h

d) RET #49
e) RET AR1
f) RET *AR1?

	4.
	What happens if we omit the instruction RET in a subroutine?

	5.
	What is meant by the term return address?

	6.
	The instruction MVMM means:

a) Move data memory to data memory

b) Move multiple memory words

c) Move memory-mapped register to memory-mapped register

d) Move multiple memory-mapped registers?

	7.
	Which two of the following are correct:

a) MVMM AR0, SP 

b) MVMM *AR0, SP

c) MVMM *AR0, *SP

d) MVMM AR4, *SP

e) MVMM ST0, AR2

f) MVMM SP, AR3

g) MVMM AR1, *AR2

	8.
	After passing parameters on the stack to a subroutine, why do we need to correct the stack pointer (SP) within the subroutine?

	9.
	The instruction FRAME +2 carries out which of the following:

a) Increments the stack pointer by 2

b) Decrements the stack pointer by 2

c) Increments auxiliary register AR2 by 2

d) Decrements auxiliary register AR2 by 2?

	10.
	What is meant by the term stack frame?

	11.
	How can room be created on the stack for local variables?

	12.
	The instruction CC means:

a) Clear Carry

b) Calculate Carry

c) Call conditionally

d) Call if Carry Clear

e) Calculate Constant

f) Compute Constant?

	13.
	Which of the following is correct:

a) CC subroutine2, BEQ

b)CC BEQ, subroutine2?

	14.
	The instruction CALA means:

a) Calculate with left shift A

b) Call Always

c) Call address in accumulator A

d) Clear accumulator A?

